
Работа 
со сводными 
таблицами
В этой глаВе...

 ● Вводный пример

 ● Создание сложных сводных таблиц

 ● Создание нескольких сводных таблиц

 ● Создание обратной сводной таблицы

Эта глава посвящена сводным таблицам Excel — одному из важнейших инструмен-
тов программы, который просто невозможно обойти стороной.

Вводный пример
Сводные таблицы — одно из самых мощных инновационных средств Excel. Они 

появились в Excel 5 и с каждой новой версией программы непрерывно улучшались. 
В этой главе вы не найдете ознакомительной информации о сводных таблицах. Пред-
полагается, что вы уже знакомы с данной темой (как и со способами создания и изме-
нения сводных таблиц вручную).

Создав сводную таблицу на основе базы данных или списка, можно быстро отобра-
зить необходимые итоговые данные, которые другим способом представить на экране 
просто невозможно. Кроме того, такое представление данных можно получить очень 
быстро. В Excel также поддерживается возможность создания кода VBA, который по-
зволяет управлять сводными таблицами.

Мы начнем с простого примера использования VBA для создания сводной табли-
цы. На рис. 8.1 показана небольшая база данных рабочего листа (диапазон ячеек). 

Она включает следующие четыре поля: Торговый представитель, Регион, Месяц и 
Продажи. Каждая запись содержит сведения об объеме продаж торгового представи-
теля за определенный месяц.

Рис. 8.1. Эта таблица является хорошим кандидатом в сводные таблицы

Файл simple pivot table.xlsm, включающий рассматриваемый далее при
мер, доступен на сайте книги.

создание сводной таблицы
На рис. 8.2 показана сводная таблица, которая создана на основе указанных дан-

ных. Она обеспечивает отображение суммарного объема ежемесячных продаж каждо-
го торгового представителя и содержит следующие поля.

➤➤ Регион. Поле фильтра отчета в сводной таблице.
➤➤ Торговый представитель. Поле строки в сводной таблице.
➤➤ Месяц. Поле столбца в сводной таблице.
➤➤ Продажи. Поле данных в сводной таблице, которое использует функцию Сумм.

Рис. 8.2. Сводная таблица, созданная на основе таблицы, показанной на рис. 8.1

Перед созданием сводной таблицы было включено средство записи макросов 
Excel 2016. Созданная сводная таблица располагается на новом рабочем листе. Ниже 
представлен автоматически сгенерированный код макроса.

Sub CreatePivotTable() 
   Sheets.Add 
   ActiveWorkbook.PivotCaches.Create _ 
      (SourceType:=xlDatabase, _ 
      SourceData:="Лист1!R1C1:R13C4", _ 
      Version:=6).CreatePivotTable _ 
      TableDestination:="Лист2!R3C1", _ 
      TableName:="PivotTable1", _ 
      DefaultVersion:=6) 
   Sheets("Лист2").Select 
   Cells(3, 1).Select 
   With ActiveSheet.PivotTables("PivotTable1") _ 
          .PivotFields("Регион") 
      .Orientation = xlPageField 
      .Position = 1 
   End With 
   With ActiveSheet.PivotTables("PivotTable1") _ 
      .PivotFields("Торговый представитель") 
      .Orientation = xlRowField 
      .Position = 1 
   End With 
   With ActiveSheet.PivotTables("PivotTable1") _ 
      .PivotFields("Месяц") 
      .Orientation = xlColumnField 
      .Position = 1 
   End With 
   ActiveSheet.PivotTables("PivotTable1") _ 
      .AddDataField ActiveSheet.PivotTables("PivotTable1") _ 
      .PivotFields("Продажи"), "Сумма по полю Продажи ", xlSum 
End Sub

Запуск записанного макроса, скорее всего, приведет к ошибке. В процессе провер-
ки кода вы обнаружите, что функция записи макроса “жестко закодировала” название 
рабочего листа (Лист2), на котором создается сводная таблица. Если подобный лист 
уже существует (либо добавляется новый лист с другим именем), выполнение макроса 
закончится неудачей. Еще более серьезной проблемой станет жестко закодированное 
имя сводной таблицы. Имя новой сводной таблицы не может быть PivotTable1, если 
в рабочей книге уже имеются другие сводные таблицы.

Несмотря на неработоспособность записанного макроса, вряд ли его можно на-
звать совсем бесполезным. Разработчики кода VBA для сводных таблиц найдут в этом 
коде немало интересного для себя.

Просмотр кода VBA
Код VBA, записанный при создании сводной таблицы, может привести вас в заме-

шательство. Для того чтобы разобраться в записанном макросе, потребуется кое-что 
знать об используемых объектах (соответствующая информация содержится в инте-
рактивном справочном руководстве).

➤➤ PivotCaches — коллекция объектов PivotCache в объекте Workbook (данные 
для сводной таблицы, которые хранятся в кеше сводной таблицы).

➤➤ PivotTables — коллекция объектов PivotTable в объекте Worksheet.
➤➤ PivotFields — коллекция полей в объекте PivotTable.
➤➤ PivotItems — коллекция отдельных элементов данных в поле.
➤➤ CreatePivotTable — метод объекта PivotCache, который создает сводную таб-
лицу на основе данных, содержащихся в кеше.

Выбор данных для сводной таблицы
Данные для сводной таблицы должны быть представлены в формате прямоугольной базы дан
ных. Причем база данных может храниться как в виде диапазона (таблица либо обычный диапа
зон), так и во внешнем файле. Несмотря на то что Excel может генерировать сводные таблицы на 
основе практически любой базы данных, должны соблюдаться определенные условия.
Поля в таблице базы данных, на основе которой создается сводная таблица, относятся к следую
щим двум типам.
•• Данные. Включает агрегируемые значения. Например, в случае с базой данных о продажах поле 
Продажи является полем данных.

•• Категория. Описывает данные. Например, в случае с той же базой данных о продажах поля Тор-
говый представитель, Регион и Месяц — это примеры полей категории, описывающих 
данные в поле Продажи.

Таблица базы данных, на основе которой может быть построена сводная таблица, должна быть 
нормализована. Иными словами, каждая запись (или строка) должна включать информацию, опи
сывающую уникальные (неповторяющиеся) данные.
Таблица базы данных может содержать произвольное количество полей данных и категорий. 
При создании сводной таблицы обычно выполняется суммирование по одному или нескольким 
полям данных. Ну а значения в полях категорий отображаются в сводной таблице в виде строк, 
столбцов либо фильтров.
Если изложенная выше концепция не вполне понятна, обратитесь к находящемуся на сайте книги 
файлу normalized data.xlsm. Эта рабочая книга содержит пример диапазона данных до и 
после выполнения нормализации, являющейся необходимым условием построения сводной та
блицы.

Улучшение кода сводной таблицы
Как и в случае с большинством записанных макросов, предыдущий пример не на-

столько эффективен, как можно ожидать. Как уже отмечалось, его выполнение навер-
няка приведет к ошибке. В принципе, его можно упростить, чтобы сделать немного 
понятнее, а также исключить возможность появления ошибок. Ниже приведен код, 
переписанный вручную, который создает ту же сводную таблицу, что и макрос из пре-
дыдущего раздела.

Sub CreatePivotTable() 
   Dim PTCache As PivotCache 
   Dim PT As PivotTable 
 
'  Создание кеша 
   Set PTCache = ActiveWorkbook.PivotCaches.Create( _ 
      SourceType:=xlDatabase, _ 
      SourceData:=Range("A1").CurrentRegion) 
 
'  Добавление нового листа для сводной таблицы 
   Worksheets.Add 
 
'  Создание сводной таблицы 
   Set PT = ActiveSheet.PivotTables.Add( _ 
      PivotCache:=PTCache, _ 
      TableDestination:=Range("A3")) 
 
'  Определение полей 
   With PT 
      .PivotFields("Регион").Orientation = xlPageField 
      .PivotFields("Месяц").Orientation = xlColumnField 
      .PivotFields("Торговый представитель") _ 
         .Orientation = xlRowField 
      .PivotFields("Продажи").Orientation = xlDataField 
 
      ' Заголовки полей отсутствуют 
      .DisplayFieldCaptions = False 
   End With 
End Sub

Процедура CreatePivotTable стала проще благодаря объявлению двух объект-
ных переменных: PTCache и PT. Новый объект PivotCache был создан с помощью 
метода Create. Также был добавлен рабочий лист, который стал активным (на этом 
листе располагается сводная таблица). Затем был создан объект PivotTable с помо-
щью метода Add из коллекции PivotTables. В последней секции кода добавляются 
четыре поля в сводную таблицу, а также задается их положение в таблице путем при-
своения значения свойству Orientation.

Заметим, что исходный макрос был жестко привязан к диапазону данных, 
на основе которого создается объект PivotCache ("Лист1!R1C1:R13C4"), и к 

местоположению сводной таблицы (Лист2). В процедуре CreatePivotTable свод-
ная таблица основана на текущем диапазоне, окружающем ячейку A1. Это гаранти-
рует, что макрос будет выполняться даже тогда, когда в диапазон будут добавлены 
дополнительные данные.

совместимость сводных таблиц
Если планируется организовать общий доступ к рабочим книгам, включающим сводные табли
цы, со стороны пользователей предыдущих версий Excel, уделите особое внимание вопросам 
совмес тимости. При просмотре кода записанного макроса (см. раздел “Создание сводной табли
цы”) нетрудно заметить следующую инструкцию:
DefaultVersion:=6

Если рабочая книга находится в режиме совместимости, на ее месте будет такая инструкция:
DefaultVersion:=xlPivotTableVersion10

Также вы отметите различия в коде записанного макроса, причина которых заключается в изме
нениях сводных таблиц начиная с версии Excel 2007.
Предположим, что в среде Excel  2016 создана рабочая книга, которая была передана пользова
телям Excel 2003. Они увидят таблицу, но не смогут ее обновить. Другими словами, они получат в 
свое распоряжение статическую таблицу, включающую ряд числовых показателей.
Для обеспечения обратной совместимости сводных таблиц в Excel 2016 следует сначала сохра
нить файл в формате XLS, а затем повторно открыть его. В таком случае гарантируется работоспо
собность сводных таблиц при открытии содержащих их документов в версиях, предшествующих 
Excel 2007. Но в этом случае вы не сможете воспользоваться новыми свойствами сводных таблиц, 
которые появились в современных версиях Excel.
К счастью, существует модуль проверки совместимости Excel, который поможет выявить пробле
мы, связанные с совместимостью. Правда, этот модуль не в состоянии проверить совместимость 
кода, генерируемого макросом. Макросы, рассматриваемые далее, не генерируют таблицы, сов
местимые со старыми версиями Excel.

Добавление рабочего листа до того, как была создана сводная таблица, исключает 
необходимость жесткого кодирования ссылки на лист. Еще одно отличие заключается 
в том, что написанный вручную макрос не определяет имя сводной таблицы. Да это-
го и не требуется, поскольку используется объектная переменная PT, которая решает 
данную задачу.

Код можно сделать более универсальным, используя индексы вместо назва
ний в коллекции PivotFields. При этом, если пользователь изменит 
заголовки столбцов, код останется работоспособным. Например, в более уни
версальном макросе следует использовать оператор PivotFields(1) вместо 
PivotFields("Регион").

Чтобы лучше понять рассматриваемую тему, запишите собственный макрос и из-
учите ключевые объекты, методы и свойства. После этого обратитесь к разделам спра-
вочного руководства, чтобы разобраться, как все это работает. Практически во всех 
случаях необходимо вносить изменения в полученный макрос. Как только вам станут 
понятны принципы управления сводными таблицами, можете приступать к созданию 
кода без предварительной записи макроса.

создание сложных сводных таблиц
В этом разделе будет представлен код VBA, используемый для создания сравни-

тельно сложной сводной таблицы.
На рис. 8.3 показан фрагмент базы данных на листе. В этой таблице содержится 

15840 строк иерархически упорядоченной информации о бюджете корпорации. Кор-
порация имеет пять филиалов, каждый филиал включает одиннадцать отделов. Отдел 
имеет четыре бюджетные категории, а каждая категория включает несколько статей 
затрат. Бюджетные и фактические расходы указываются для каждого (из двенадцати) 
месяцев. Цель — создать сводную таблицу, которая суммирует эту информацию.

Рис. 8.3. Данные из этой рабочей книги будут собраны в сводной таблице

Файл budget pivot table.xlsm, включающий рассматриваемый в данном 
разделе пример, доступен на сайте книги.

На рис. 8.4 показана сводная таблица, созданная на основе приведенных выше дан-
ных. Обратите внимание на то, что она содержит вычисляемое поле, которое называ-
ется Отклонение. Значение этого поля представляет собой разницу между значения-
ми полей План и Факт.

Рис. 8.4. Сводная таблица, созданная на основе данных из рис. 8.3

Можно также вставить новую колонку в таблицу и создать формулу, вычисляю
щую разницу между плановыми и фактическими показателями. Эта возможность 
будет недоступной в том случае, если данные берутся из внешнего источника.

Код сводной таблицы
Ниже приведен код, генерирующий сводную таблицу.

Sub CreatePivotTable() 
   Dim PTcache As PivotCache 
   Dim PT As PivotTable 
 
   Application.ScreenUpdating = False 
'  Удаление листа сводной таблицы (в случае его наличия) 
   On Error Resume Next 
   Application.DisplayAlerts = False 

09_ch08.indd   336 30.10.2017   23:03:46



глава 8. Работа со сводными таблицами 337

   Sheets("Сводная таблица").Delete 
      On Error GoTo 0 
 
'  Создание кеша сводной таблицы  
   Set PTcache = ActiveWorkbook.PivotCaches.Create( _ 
      SourceType:=xlDatabase, _ 
      SourceData:=Range("A1").CurrentRegion.Address) 
 
'  Добавление нового рабочего листа 
   Worksheets.Add 
   ActiveSheet.Name = "Сводная таблица" 
   ActiveWindow.DisplayGridlines = False 
 
'  Создание сводной таблицы на основе данных кеша 
   Set PT = ActiveSheet.PivotTables.Add( _ 
      PivotCache:=PTcache, _ 
      TableDestination:=Range("A1"), _ 
      TableName:="BudgetPivot") 
 
   With PT 
'     Добавление полей 
      .PivotFields("Категория").Orientation = xlPageField 
      .PivotFields("Филиал").Orientation = xlPageField 
      .PivotFields("Отдел").Orientation = xlRowField 
      .PivotFields("Месяц").Orientation = xlColumnField 
      .PivotFields("План").Orientation = xlDataField 
      .PivotFields("Факт").Orientation = xlDataField 
      .DataPivotField.Orientation = xlRowField 
 
'     Добавление вычисляемого поля, определяющего отклонение 
      .CalculatedFields.Add "Отклонение", "=План-Факт" 
      .PivotFields("Отклонение").Orientation = xlDataField 
 
'     Определение числового формата 
      .DataBodyRange.NumberFormat = "0,000" 
 
'     Применение стиля 
      .TableStyle2 = "PivotStyleMedium2" 
 
'     Сокрытие заголовков полей 
      .DisplayFieldCaptions = False 
 
'     Изменение заголовков 
      .PivotFields("Сумма по полю План").Caption = " План" 
      .PivotFields("Сумма по полю Факт").Caption = " Факт" 
      .PivotFields("Сумма по полю Отклонение").Caption = _ 
         " Отклонение" 
   End With 
End Sub

Принцип работы сводной таблицы
Процедура CreatePivotTable начинает свою работу с удаления листа Сводная 

таблица, если он существует. После этого создается объект PivotCache, добавля-
ется новый лист Сводная таблица и создается сводная таблица на основе объекта 
PivotCache. Далее программа добавляет поля к созданной сводной таблице:

➤➤ Категория — поле фильтра отчета (страницы);
➤➤ Филиал — поле фильтра отчета (страницы);
➤➤ Отдел — поле строки;
➤➤ Месяц — поле столбца;
➤➤ План — поле данных;
➤➤ Факт — поле данных.

Обратите внимание на то, что свойство Orientation объекта DataPivotField 
устанавливается равным xlRowField с помощью следующего оператора:

.DataPivotField.Orientation = xlRowField

Этот оператор определяет общую ориентацию сводной таблицы и представляет 
поле в области ЗНАЧЕНИЯ в списке полей сводной таблицы (рис. 8.5). Попытайтесь 
переместить это поле в область СТОЛБЦЫ, после чего посмотрите, как изменится ма-
кет сводной таблицы.

Затем процедура использует метод Add из коллекции CalculatedFields для соз-
дания вычисляемого поля Отклонение, величина которого представляет собой резуль-
тат вычитания значения поля Факт из значения поля План. Это вычисляемое поле 
относится к категории полей данных.

Для добавления вычисляемого поля в сводную таблицу вручную воспользуйтесь 
командой Работа со сводными таблицамиðАнализðВычисленияðПоля, 
элементы и наборыðВычисляемое поле, чтобы перейти в диалоговое окно 
Вставка вычисляемого поля.

Также код выполняет небольшие “косметические” улучшения сводной таблицы:

➤➤ применяет числовой формат к объекту DataBodyRange (этот объект представляет 
данные всей сводной таблицы);

➤➤ применяет стиль;
➤➤ скрывает заголовки (эквивалентен элементу управления Работа со сводными таб
лицамиðАнализðПоказать/СкрытьðЗаголовки полей;

➤➤ изменяет заголовки, отображенные в сводной таблице. Например, название Сумма 
по полю План заменяется названием " План". Заметьте, что перед словом План 
вставляется пробел. В Excel не допускается изменение заголовка, который соответ-
ствует имени поля, но, как видите, это ограничение можно обойти, добавив пробел.

Рис. 8.5. Список полей сводной таблицы

Не забывайте пользоваться средством записи макросов для выявления требуе
мых элементов VBA. Выполняйте различные действия и смотрите, какой код при 
этом генерируется. В сочетании с информацией из справочной системы средство 
записи макросов даст вам все необходимое для написания собственных про
грамм (коечто, конечно, придется выяснять методом проб и ошибок).

создание нескольких сводных таблиц
При выполнении данного примера создается целая серия сводных таблиц, в 

которых суммируются данные, собранные в процессе опроса покупателей. Эти данные 
находятся в базе данных, которая состоит из 150 строк. Каждая строка включает 
указание пола респондента, а также числовой рейтинг, который изменяется от 1 до 5 
по каждому из 14 пунктов опроса.

Этот пример доступен на сайте книги (файл survey data pivot tables.
xlsm).

На рис. 8.6 показаны некоторые из 28 сводных таблиц, созданные макросом. Сум-
мирование по каждому элементу опроса производилось с помощью двух сводных та-
блиц (в одной отображаются фактические результаты опроса, в другой — процент 
респондентов, ответивших на пункты опроса).

Рис. 8.6. Некоторые из 28 сводных таблиц, созданных с помощью процедуры VBA

Ниже приводится код процедуры VBA, с помощью которого создавались сводные 
таблицы.

Sub MakePivotTables() 
'  Эта процедура создает 28 сводных таблиц 
   Dim PTCache As PivotCache 
   Dim PT As PivotTable 
   Dim SummarySheet As Worksheet 
   Dim ItemName As String 
   Dim Row As Long, Col As Long, i As Long 
 
   Application.ScreenUpdating = False 
 
'  Удаление листа "Сводка" при его наличии 
   On Error Resume Next 
   Application.DisplayAlerts = False 
   Sheets("Сводка").Delete 
   On Error GoTo 0 
 
'  Добавление листа "Сводка" 
   Set SummarySheet = Worksheets.Add 
   ActiveSheet.Name = "Сводка" 

'  Создание кеша сводной таблицы 
   Set PTCache = ActiveWorkbook.PivotCaches.Create( _ 
      SourceType:=xlDatabase, _ 
      SourceData:=Sheets("ДанныеОпроса").Range("A1"). _ 
      CurrentRegion) 
 
   Row = 1 
      For i = 1 To 14 
         For Col = 1 To 6 Step 5    ' 2 столбца 
            ItemName = Sheets("ДанныеОпроса").Cells(1, i + 2) 
            With Cells(Row, Col) 
               .Value = ItemName 
               .Font.Size = 16 
            End With 
 
'           Создание сводной таблицы 
            Set PT = ActiveSheet.PivotTables.Add( _ 
               PivotCache:=PTCache, _ 
            TableDestination:=SummarySheet.Cells( _ 
               Row + 1, Col)) 
 
'           Добавление полей 
            If Col = 1 Then    ' таблицы частот 
               With PT.PivotFields(ItemName) 
                  .Orientation = xlDataField 
                  .Name = "Частота" 
                  .Function = xlCount 
               End With 
            Else     ' процентные таблицы 
               With PT.PivotFields(ItemName) 
                  .Orientation = xlDataField 
                  .Name = "Процент" 
                  .Function = xlCount 
                  .Calculation = xlPercentOfColumn 
                  .NumberFormat = "0,0%" 
               End With 
            End If 
 
            PT.PivotFields(ItemName).Orientation = xlRowField 
            PT.PivotFields("Пол").Orientation = xlColumnField 
            PT.TableStyle2 = "PivotStyleMedium2" 
            PT.DisplayFieldCaptions = False 
            If Col = 6 Then 
'              Добавление шкал данных в последний столбец 
               PT.ColumnGrand = False 
               PT.DataBodyRange.Columns(3).FormatConditions. _ 
                  AddDatabar 
               With pt.DataBodyRange.Columns(3). _ 

                  FormatConditions(1) 
                  .BarFillType = xlDataBarFillSolid 
                  .MinPoint.Modify newtype:= _ 
                      xlConditionValueNumber, newvalue:=0 
                  .MaxPoint.Modify newtype:= _ 
                     xlConditionValueNumber, newvalue:=1 
               End With 
            End If 
         Next Col 
         Row = Row + 10 
      Next i 
 
'     Замена чисел описательным текстом 
      With Range("A:A,F:F") 
         .Replace "1", "Категорически не согласен" 
         .Replace "2", "Не согласен" 
         .Replace "3", "Не определился" 
         .Replace "4", "Согласен" 
         .Replace "5", "Полностью согласен" 
      End With 
End Sub

Заметим, что все сводные таблицы были созданы на основе единственного объекта 
PivotCache.

Сводные таблицы создаются во вложенных циклах. Значение счетчика цикла Col 
изменяется от 1 до 6 с помощью параметра Step. Для второго столбца сводных таблиц 
немного изменяется характер обработки. Выполняются следующие действия:

➤➤ отображается величина (в виде процента от значения в столбце);
➤➤ не отображаются окончательные итоги для строк;
➤➤ используется числовой формат;
➤➤ отображаются цветовые шкалы, с помощью которых реализуется условное форма-
тирование.

Переменная Row отслеживает начальную строку в каждой сводной таблице. Завер-
шающая операция заключается в замене числовых категорий в столбцах A и F текстом. 
Например, вместо единицы подставляется текст “Категорически не согласен”.

создание обратной сводной таблицы
Сводная таблица представляет собой результат суммирования данных в обычной 

таблице. А как быть в случае, если у вас имеется итоговая таблица, на основе кото-
рой нужно воссоздать исходную таблицу? Соответствующий пример приводится на 
рис. 8.7. Диапазон ячеек B2:F14 представляет собой итоговую таблицу — упрощенный 
вариант сводной таблицы. В столбцах I:K находится состоящая из 48 строк таблица, 

созданная на основе итоговой таблицы. В этой таблице каждая строка содержит точку 
данных, а первые два столбца включают описание этой точки.

Рис. 8.7. Итоговая таблица (слева) может быть преобразована в обычную 
таблицу (справа)

В Excel не существует прямого способа преобразования итоговой таблицы в обыч-
ную, но эту работу поможет выполнить макрос VBA. После его создания можно 
разработать диалоговое окно UserForm, показанное на рис. 8.8. В нем определяются 
входной и выходной диапазоны, а также имеется опция преобразования выходного 
диапазона в таблицу.

Рис. 8.8. В этом диалоговом окне пользователь 
может определить диапазоны

Этот пример доступен на сайте книги (файл reverse pivot table.xlsm).

После щелчка мышью на кнопке OK в диалоговом окне UserForm код VBA прове-
ряет диапазоны и вызывает процедуру ReversePivot с помощью следующего опе-
ратора:

Call ReversePivot(SummaryTable, OutputRange, CreateTable)

Этой процедуре передаются следующие три аргумента.

➤➤ SummaryTable. Объект Range, который представляет итоговую таблицу.
➤➤ OutputRange. Объект Range, представляющий левую верхнюю ячейку выходного 
диапазона.

➤➤ CreateTable. Флажок (объект Checkbox) в окне UserForm.

Эта процедура работает с итоговой таблицей любого размера. Количество строк 
данных в выходной таблице можно подсчитать по формуле (r-1) * (c-1), в которой 
переменные r и c представляют собой количество строк и столбцов в итоговой табли-
це соответственно.

Ниже приведен код процедуры ReversePivot.
Sub ReversePivot(SummaryTable As Range, _ 
   OutputRange As Range, CreateTable As Boolean) 
   Dim r As Long, c As Long 
   Dim OutRow As Long, OutCol As Long 
 
'  Преобразование диапазона 
   OutRow = 2 
   Application.ScreenUpdating = False 
   OutputRange.Range("A1:C3") = Array("Столбец1", _ 
      "Столбец2", "Столбец3") 
   For r = 2 To SummaryTable.Rows.Count 
      For c = 2 To SummaryTable.Columns.Count 
         OutputRange.Cells(OutRow, 1) = SummaryTable.Cells(r, 1) 
         OutputRange.Cells(OutRow, 2) = SummaryTable.Cells(1, c) 
         OutputRange.Cells(OutRow, 3) = SummaryTable.Cells(r, c) 
         OutRow = OutRow + 1 
      Next c 
   Next r 
 
'  Создать таблицу? 
   If CreateTable Then _ 
      ActiveSheet.ListObjects.Add xlSrcRange, _ 
         OutputRange.CurrentRegion, , xlYes 
End Sub

Описанная процедура достаточно проста. Код выполняет циклический обход строк 
и столбцов во входном диапазоне, а затем записывает данные в выходной диапазон, 
который всегда состоит из трех столбцов. Переменная OutRow отслеживает текущую 
строку в выходном диапазоне. Если пользователь устанавливает флажок, выходной 
диапазон преобразуется в таблицу. При этом используется метод Add из коллекции 
ListObjects.

